

I217: Functional Programming

6. Infinite Lists

Kazuhiro Ogata

Roadmap


- E-strategy
- Infinite Lists
- Sieve of Eratosthenes
- Hamming's Problem
- Simulator of a Mutex Protocol

E-strategy

```

mod! NAT-TRIPLE-OMLM { pr(NAT) [NatTriple]
  op (_,_,_) : Nat Nat Nat -> NatTriple {constr} .
  op 1st : NatTriple -> Nat {strat: (0 1 0)} .
  vars FE SE TE : Nat .
  eq 1st((FE,SE,TE)) = FE . }

```


the local strategy

```

set trace whole on
open NAT-TRIPLE-OMLM .
  red 1st((1 + 1,2 + 2,3 + 3)) .
close
set trace whole off

```

```

-- reduce in %NAT-TRIPLE-OMLM : (1st(((1 + 1) , (2 + 2) , (3 + 3)))):Nat
[1]: (1st(((1 + 1) , (2 + 2) , (3 + 3)))):Nat
---> (1 + 1):NzNat
[2]: (1 + 1):NzNat
---> (2):NzNat
(2):NzNat

```

[1] Since the given term is a redex, one-step rewrite is performed.

[2] 1 + 1 is rewritten to 2.

E-strategy

```

op 1st : NatTriple -> Nat {strat: (1 0)} .

```

The local strategy (1 0) given to the operator 1st says that when reducing a ground term $1st(arg)$, CafeOBJ first reduces the 1st argument arg to arg' and performs one-step rewrite to $1st(arg')$ if $1st(arg')$ is a redex; otherwise CafeOBJ returns $1st(arg)$ as the result; after the one-step rewrite CafeOBJ reduces the contract based on the local strategy given to the top operator of the contract.

```

op 1st : NatTriple -> Nat {strat: (0 1 0)} .

```

(0 1 0) says that when reducing a ground term $1st(arg)$, CafeOBJ first performs one-step rewrite to $1st(arg)$ if $1st(arg)$ is a redex; otherwise CafeOBJ does the same as it does for the local strategy (1 0); after the one-step rewrite CafeOBJ reduces the contract based on the local strategy given to the top operator of the contract.

E-strategy

The reduction strategy adopted by CafeOBJ is *E-strategy* that selects a redex based on local strategies given to operators as follows.

$$\mathbf{op} f : S_1 \dots S_n \rightarrow S \text{ \textit{the local strategy} } \{ \mathbf{strat} : (x_1 \dots x_m) \} .$$

where $0 \leq x_i \leq n$ for each $i = 1, \dots, m$.

When reducing $f(a_1, \dots, a_n)$ referred as t , Note that t may be modified in the following.
for $j = x_1, \dots, x_m$,

(1) if $j = 0$, CafeOBJ performs one-step rewrite if t is a redex, reduces the contract and returns the result of reducing the contract as the result;

(2) otherwise, CafeOBJ reduces t_j ;

CafeOBJ returns t as the result.

(Note that the result may contain redexes.) Note that t may not be the same as $f(a_1, \dots, a_n)$.

E-strategy

Operators can be given local strategies by human users.

If human users do not so, the CafeOBJ system give some adequate local strategies to operators.

```

mod! NAT-IF { pr(NAT)
  op if_then{ }else{ } : Bool Nat Nat -> Nat .
  vars N1 N2 : Nat .
  eq if true then {N1} else {N2} = N1 .
  eq if false then {N1} else {N2} = N2 . }

```

The CafeOBJ system gives (1 0 2 3) to `if_then{ }else{ }` as the local strategy.

```
show NAT-IF
```

Infinite Lists

Lists that consist of an infinite number of elements

Based on the following

```

op nil : -> Nil {constr} .
op _|_ : Elt.E List -> NnList {constr} .

```

any lists that consist of an arbitrary finite number of elements can be constructed but any infinite lists cannot.

Infinite Lists

One way to describe infinite lists in CafeOBJ is as follows:

```

mod* INF-LIST(E :: TRIV) {
  [InfList]
  op _|_ : Elt.E InfList -> InfList {strat: (1 0)} .
}

```

Note that this local strategy is given to _|_

$e_1 | e_2 | \dots | e_n | il$ is a term of InfList if e_1, e_2, \dots, e_n are terms of Elt.E and il is a term of InfList.

```

mod! GLIST(E :: TRIV) {
  [Nil NnList < List]
  op nil : -> Nil {constr} .
  op _|_ : Elt.E List -> List {constr} .
}

```

The module INF-LIST imports the modules GLIST and NAT:

```

pr(NAT)
pr(GLIST(E))

```

Infinite Lists

Some functions for infinite lists:

```
op take : InfList Nat -> List .
eq take(IL,0) = nil .
eq take(X | IL, NzN) = X | take(IL,p NzN) .
```

```
op drop : InfList Nat -> InfList .
eq drop(IL,0) = IL .
eq drop(X | IL, NzN) = drop(IL,p NzN) .
```

Note that the following variables are declared in INF-LIST:

```
vars X Y : Elt.E .
vars IL IL2 : InfList .
var NzN : NzNat .
var N : Nat .
var L : List .
```

Infinite Lists

```
op _@_ : List InfList -> InfList .
eq nil @ IL = IL .
eq (X | L) @ IL = X | (L @ IL) .
```

```
op zip : InfList InfList -> InfList .
eq zip(X | IL, Y | IL2) = X | Y | zip(IL,IL2) .
```


Sieve of Eratosthenes

```
open ERATOSTHENES-SIEVE .  
  red primes .  
  red take(primes,10) .  
  red take(primes,20) .  
  red take(primes,50) .  
  red take(primes,100) .  
close
```

Hamming's Problem

```
mod! HAMMING {  
  pr(NAT-INF-LIST)  
  op ham : -> InfList .  
  op 2* : InfList -> InfList .  
  op 3* : InfList -> InfList .  
  op 5* : InfList -> InfList .  
  op merge : InfList InfList -> InfList .  
  vars X Y : Nat .  
  vars IL IL2 : InfList .  
  -- ham  
  eq ham = 1 | merge(merge(2*(ham),3*(ham)),5*(ham)) .  
}
```


Hamming's Problem

```

-- 2*
eq 2*(X | IL) = 2 * X | 2*(IL) .
-- 3*
eq 3*(X | IL) = 3 * X | 3*(IL) .
-- 5*
eq 5*(X | IL) = 5 * X | 5*(IL) .
-- merge
eq merge(X | IL, Y | IL2)
  = if X < Y
    then {X | merge(IL, Y | IL2)}
    else {if Y < X
          then {Y | merge(X | IL, IL2)}
          else {X | merge(IL, IL2)} } .
}

```

Hamming's Problem

```

open HAMMING .
  red ham .
  red take(ham,10) .
  red take(ham,20) .
  red take(ham,50) .
  red take(ham,100) .
close

```

Simulator of a Mutex Protocol

Let us consider a multi-threaded program in which multi-threads are running simultaneously and sharing some resources, such as objects.

Some shared resources must be used mutually exclusively by at most one thread at any given moment.

A mechanism to achieve this is called a *mutual exclusion (Mutex) protocol*.

Simulator of a Mutex Protocol

Suppose that there are two threads t_1 & t_2 that share a Boolean variable *locked* whose initial value is *false* and execute the following pseudo-code:

<p>Each thread does something that requires some shared resources in "Critical Section".</p>	<p>Loop: "Remainder Section" rs: while <i>locked</i> = true { ms: <i>locked</i> := true; "Critical Section" cs: <i>locked</i> := false;</p>	<p>Each thread does something that does not require any shared resources in "Remainder Section".</p>
--	---	--

Each thread t is at rs, ms or cs. When t is at rs, it can check if *locked* is false. If so, it moves to ms and sets *locked* true, entering "Critical Section" and doing something that requires some shared resources. Otherwise, it stays at rs. When t is at cs, it sets *locked* false, going back to "Remainder Section".

Simulator of a Mutex Protocol

Each state of the protocol is characterized by three values: *locked*, the location of t1 and the location of t2. Therefore, a state of the protocol is expressed as

$$(\text{locked}: b, \text{pc1}: l_1, \text{pc2}: l_2)$$

where b is a Boolean value and l_1 and l_2 are rs, ms or cs.

```

mod! STATE principal-sort State {
  pr(LOC)
  [State]
  op (locked: _,pc1: _,pc2: _) : Bool Loc Loc -> State {constr} .
}

```

Simulator of a Mutex Protocol

```

mod! TID {
  [Tid]
  ops t1 t2 : -> Tid {constr} .
  op if_then{__}else{__} : Bool Tid Tid -> Tid .
  vars T1 T2 : Tid .
  eq if true then {T1} else {T2} = T1 .
  eq if false then {T1} else {T2} = T2 .
}

mod! LOC {
  [Loc]
  ops rs ms cs : -> Loc {constr} .
}

```

Simulator of a Mutex Protocol

Given a state and a thread, the next state can be determined.

(locked: false, pc1: rs, pc2: rs)

t1



(locked: false, pc1: ms, pc2: rs)

(locked: false, pc1: ms, pc2: rs)

t2



(locked: false, pc1: ms, pc2: ms)

(locked: false, pc1: ms, pc2: ms)

t2



(locked: true, pc1: ms, pc2: cs)

Simulator of a Mutex Protocol

```

mod! FMUTEX { pr(STATE) pr(TID)
op trans : State Tid -> State .
vars L1 L2 : Loc . var B : Bool .
-- t1
eq trans((locked: true,pc1: rs,pc2: L2),t1)
  = (locked: true,pc1: rs,pc2: L2) .
eq trans((locked: false,pc1: rs,pc2: L2),t1)
  = (locked: false,pc1: ms,pc2: L2) .
eq trans((locked: B,pc1: ms,pc2: L2),t1)
  = (locked: true,pc1: cs,pc2: L2) .
eq trans((locked: B,pc1: cs,pc2: L2),t1)
  = (locked: false,pc1: rs,pc2: L2) .

```

Simulator of a Mutex Protocol

```

-- t2
eq trans((locked: true,pc1: L1,pc2: rs),t2)
  = (locked: true,pc1: L1,pc2: rs) .
eq trans((locked: false,pc1: L1,pc2: rs),t2)
  = (locked: false,pc1: L1,pc2: ms) .
eq trans((locked: B,pc1: L1,pc2: ms),t2)
  = (locked: true,pc1: L1,pc2: cs) .
eq trans((locked: B,pc1: L1,pc2: cs),t2)
  = (locked: false,pc1: L1,pc2: rs) .
}

```

Simulator of a Mutex Protocol

The simulator takes a state and an infinite list of thread IDs (called a scheduling), and generates an infinite list of states (called a computation).

```

mod! COMP {
  pr(INF-LIST(STATE)
    * {sort InfList -> Comp, sort List -> FComp} )
}

mod! SCHED { pr(NAT)
  pr(INF-LIST(TID) * {sort InfList -> Sched} )
  op sched : Nat -> Sched .
  var N : Nat .
  eq sched(N) = if 2 divides N
    then {t1 | sched(N quo 2)} else {t2 | sched(N quo 2)} .
}

```

n of sched(*n*) determines when to choose which threads, namely the scheduling.

Simulator of a Mutex Protocol

```

open SCHED .
  red take(sched(123),10) .
  red take(sched(1234),10) .
  red take(sched(12345),10) .
close

```

Simulator of a Mutex Protocol

```

mod! SIM { pr(FMUTEX) pr(COMP) pr(SCHED)
  op sim : State Nat -> Comp .
  op sub-sim : State Sched -> Comp .
  var S : State . var N : Nat . var NzD : NzNat .
  var T : Tid . var TIL : Sched .
  eq sim(S,N) = sub-sim(S,sched(N)) .
  eq sub-sim(S,T | TIL) = S | sub-sim(trans(S,T),TIL) .
}

open SIM .
  red take(sim((locked: false,pc1: rs,pc2: rs),123),10) .
  red take(sim((locked: false,pc1: rs,pc2: rs),1234),10) .
  red take(sim((locked: false,pc1: rs,pc2: rs),12345),10) .
close

```

Simulator of a Mutex Protocol

One desired property mutex protocols should satisfy is called the *mutex property* that there is at most one thread in “Critical Section” at any given moment. The simulator is revised so that it can check in a specified depth if each state generated satisfies the property. If the simulator finds a state in which the property is broken, it returns the computation fragment leading to the state.

The property is defined as follows:

```

op mutex : State -> Bool .
vars L1 L2 : Loc .
var B : Bool .
eq mutex((locked: B,pc1: L1,pc2: L2)) = not (L1 == cs and L2 == cs) .

```

Simulator of a Mutex Protocol

The revised simulator is as follows:

```

op sim-check : State Nat Nat -> FComp .
op sub-sim-check : State Sched Nat -> FComp .
var D : Nat . var NzD : NzNat .
eq sim-check(S,N,D) = sub-sim-check(S,sched(N),D) .
eq sub-sim-check(S,T | TIL,0) = S | nil .
eq sub-sim-check(S,T | TIL,NzD)
  = if mutex(S) then {S | sub-sim-check(trans(S,T),TIL,p NzD)}
    else {S | nil} .

open SIM .
  red sim-check((locked: false,pc1: rs,pc2: rs),123,10) .
  red sim-check((locked: false,pc1: rs,pc2: rs),1234,10) .
  red sim-check((locked: false,pc1: rs,pc2: rs),12345,10) .
close

```

As you can see, the protocol does not satisfy the property. Please see Appendix for a correct version of the protocol.

Exercises

1. Write all programs in the slides and feed them into the CafeOBJ system. Moreover, write some more test code and do some more testing for the programs.
2. Revise the simulator (including the revised one) so that it can deal with the case in which there are four threads.

Appendix

A correct version of the protocol:

```
Loop: "Remainder Section"  
rs: while test&set(locked) = true {}  
   "Critical Section"  
cs: locked := false;
```

test&set(*x*) does the following atomically:

```
tmp := x;  
x := true;  
return tmp;
```